Basic Statistical Tests Using SAS
(commands= finan_basic_stat.sas)
.The examples in this handout use the PULSE data, which is available in the labdata folder as an Excel file (PULSE.XLS), as a raw data file (PULSE.DAT), and as a CSV file (PULSE.CSV). Information on the variables in the PULSE data is included in the handouts from day 1.
Read in the raw data for PULSE:
/*Specify generic output using Options Formchar= */

options formchar="|----|+|---+=|-/\<>*";

/*Read in the raw data. Set up variable labels.*/

data pulse;

   infile "C:\temp\labdata\pulse.csv" firstobs=2 delimiter="," missover;

   input pulse1 pulse2 ran smokes sex height weight activity;

run;

/*Print out the first 25 cases*/

proc print data=pulse (obs=25);

run;

Create and assign formats to variables:
/*Create temporary formats and assign formats to corresponding variables*/

proc format;


value sexfmt 1="Male" 2="Female";


value yesnofmt  1="Yes" 2="No";


value actfmt 1="Low" 2="Medium" 3="High";

run;

data pulse;


set pulse;


format sex sexfmt. ran smokes yesnofmt. activity actfmt.; 


label pulse1 = "Resting pulse, rate per minute"



pulse2 = "Second pulse, rate per minute";

run;

proc print data=pulse (obs=25) label;

run;
Output from these commands is shown below:
          Resting         Second

        pulse, rate    pulse, rate

 Obs     per minute     per minute    ran    smokes     sex      height    weight    activity

   1         64             88        Yes     No       Male        66        140      Medium

   2         58             70        Yes     No       Male        72        145      Medium

   3         62             76        Yes     Yes      Male        73        160      High

   4         66             78        Yes     Yes      Male        73        190      Low

   5         64             80        Yes     No       Male        69        155      Medium

   6         74             84        Yes     No       Male        73        165      Low

   7         84             84        Yes     No       Male        72        150      High

   8         68             72        Yes     No       Male        74        190      Medium

   9         62             75        Yes     No       Male        72        195      Medium

  10         76            118        Yes     No       Male        71        138      Medium

  11         90             94        Yes     Yes      Male        74        160      Low

  12         80             96        Yes     No       Male        72        155      Medium

  13         92             84        Yes     Yes      Male        70        153      High

  14         68             76        Yes     No       Male        67        145      Medium

  15         60             76        Yes     No       Male        71        170      High

  16         62             58        Yes     No       Male        72        175      High

  17         66             82        Yes     Yes      Male        69        175      Medium

  18         70             72        Yes     Yes      Male        73        170      High

  19         68             76        Yes     Yes      Male        74        180      Medium

  20         72             80        Yes     No       Male        66        135      High

  21         70            106        Yes     No       Male        71        170      Medium

  22         74             76        Yes     No       Male        70        157      Medium

  23         66            102        Yes     No       Male        70        130      Medium

  24         70             94        Yes     Yes      Male        75        185      Medium

  25         96            140        Yes     No       Female      61        140      Medium
/*Display data file information and variable lists */

proc contents data=pulse varnum;

run;

/*Obtain descriptive statistics */

proc means data=pulse; 

run;
                                         The MEANS Procedure

 Variable  Label                            N          Mean       Std Dev       Minimum       Maximum

 ----------------------------------------------------------------------------------------------------

 pulse1    Resting pulse, rate per minute  92    72.8695652    11.0087052    48.0000000   100.0000000

 pulse2    Second pulse, rate per minute   92    80.0000000    17.0937943    50.0000000   140.0000000

 ran                                       92     1.6195652     0.4881540     1.0000000     2.0000000

 smokes                                    92     1.6956522     0.4626519     1.0000000     2.0000000

 sex                                       92     1.3804348     0.4881540     1.0000000     2.0000000

 height                                    92    68.7391304     3.6520943    61.0000000    75.0000000

 weight                                    92   145.1521739    23.7393978    95.0000000   215.0000000

 activity                                  92     2.1195652     0.5711448     1.0000000     3.0000000

 ----------------------------------------------------------------------------------------------------
One-sample test for a categorical variable:
You can test whether the proportions in given levels of a categorical fit hypothesized values by using a chi-square goodness of fit test. Specify the proportions you wish to test by using the testp= option in the tables statement. This allows you to specify any proportions that you wish for each level of the test variable. But the proportions have to add up to either 100 or 1.00.
/*One-sample Chi-square test: comparing the sample distribution againST a user-specified distribution*/

/*Test variable has two levels*/

proc freq data=pulse;


tables sex /chisq testp=(55, 45) ;

run;

                                          The FREQ Procedure

                                                     Test     Cumulative    Cumulative

                  sex    Frequency     Percent     Percent     Frequency      Percent

               -----------------------------------------------------------------------

               Male            57       61.96       55.00            57        61.96

               Female          35       38.04       45.00            92       100.00

                                           Chi-Square Test

                                      for Specified Proportions

                                      -------------------------

                                      Chi-Square         1.7989

                                      DF                      1

                                      Pr > ChiSq         0.1799

                                           Sample Size = 92

/*You can do the test for a subset of the data*/
title "Binomial test of sex distribution for smokers";

proc freq data=pulse;


tables sex / chisq testp=(80, 20);


where smokes=1;

run;

title;
                            Binomial test of sex distribution for smokers                           

                                          The FREQ Procedure

                                                     Test     Cumulative    Cumulative

                  sex    Frequency     Percent     Percent     Frequency      Percent

               -----------------------------------------------------------------------

               Male            20       71.43       80.00            20        71.43

               Female           8       28.57       20.00            28       100.00

                                           Chi-Square Test

                                      for Specified Proportions

                                      -------------------------

                                      Chi-Square         1.2857

                                      DF                      1

                                      Pr > ChiSq         0.2568

                                           Sample Size = 28

The test categorical variable could have more than two levels.
/*Test variable has more than two levels*/

proc freq data=pulse;


tables activity /chisq testp=(.20, .50, .30);

run;

                                          The FREQ Procedure

                                                      Test     Cumulative    Cumulative

              activity    Frequency     Percent     Percent     Frequency      Percent

              -------------------------------------------------------------------------

                Low             10       10.87       20.00            10        10.87

                Medium          61       66.30       50.00            71        77.17

                High            21       22.83       30.00            92       100.00

                                           Chi-Square Test

                                      for Specified Proportions

                                      -------------------------

                                      Chi-Square        10.3043

                                      DF                      2

                                      Pr > ChiSq         0.0058

                                           Sample Size = 92

One-Sample t test
You can use Proc Univariate to test the population mean against any null hypothesis value you specify, by using mu0= option. Equivalently, you can carry out a one-sample t-test in Proc Ttest by specifying the H0= option.
/*One-sample T test*/

proc univariate data=pulse mu0=72;


var pulse1;


histogram / normal (mu=est sigma=est);


qqplot /normal (mu=est sigma=est);

run;
Selected output from Proc Univariate:
                                      Proc Univariate

                                  Tests for Location: Mu0=72

                           Test           -Statistic-    -----p Value------

                           Student's t    t  0.757635    Pr > |t|    0.4506

                           Sign           M        -3    Pr >= |M|   0.5900

                           Signed Rank    S      96.5    Pr >= |S|   0.6797

proc ttest data=pulse H0=72 ;

  var pulse1;

run;

                                      The TTEST Procedure

                                          Statistics

                      Lower CL           Upper CL   Lower CL             Upper CL

 Variable       N       Mean     Mean       Mean    Std Dev   Std Dev    Std Dev   StdErr

 pulse1        92      70.59    72.87     75.149     9.6155    11.009     12.878   1.1477

                                            T-Tests

                            Variable      DF    t Value    Pr > |t|

                            pulse1        91       0.76      0.4506
Associations between two categorical variables:

If you wish to examine the association between two categorical variables, you can use Proc Freq to obtain a contingency table, which is also known as a cross-tabulation. The appropriate tests are a chi-square test or Fisher’s exact test.
/*Test of independence between two categorical variables */

proc freq data=pulse;


tables sex * activity / nopercent nocol expected chisq;

run;

                                     The FREQ Procedure

                                       Table of sex by activity

                             sex       activity

                             Frequency|

                             Expected |

                             Row Pct  |Low     |Medium  |High    |  Total

                             ---------+--------+--------+--------+

                             Male     |      6 |     35 |     16 |     57

                                      | 6.1957 | 37.793 | 13.011 |

                                      |  10.53 |  61.40 |  28.07 |

                             ---------+--------+--------+--------+

                             Female   |      4 |     26 |      5 |     35

                                      | 3.8043 | 23.207 | 7.9891 |

                                      |  11.43 |  74.29 |  14.29 |

                             ---------+--------+--------+--------+

                             Total          10       61       21       92

                               Statistics for Table of sex by activity

                        Statistic                     DF       Value      Prob

                        ------------------------------------------------------

                        Chi-Square                     2      2.3641    0.3067

                        Likelihood Ratio Chi-Square    2      2.4827    0.2890

                        Mantel-Haenszel Chi-Square     1      1.4339    0.2311

                        Phi Coefficient                       0.1603

                        Contingency Coefficient               0.1583

                        Cramer's V                            0.1603

                                           Sample Size = 92

/*SAME TEST, BUT for non-smokers only*/
proc freq data=pulse;


tables sex * activity /nopercent nocol expected chisq;


where smokes=2;

run;

                                          The FREQ Procedure

                                       Table of sex by activity

                             sex       activity

                             Frequency|

                             Expected |

                             Row Pct  |Low     |Medium  |High    |  Total

                             ---------+--------+--------+--------+

                             Male     |      3 |     22 |     12 |     37

                                      | 3.4688 | 24.281 |   9.25 |

                                      |   8.11 |  59.46 |  32.43 |

                             ---------+--------+--------+--------+

                             Female   |      3 |     20 |      4 |     27

                                      | 2.5313 | 17.719 |   6.75 |

                                      |  11.11 |  74.07 |  14.81 |

                             ---------+--------+--------+--------+

                             Total           6       42       16       64

                               Statistics for Table of sex by activity

                        Statistic                     DF       Value      Prob

                        ------------------------------------------------------

                        Chi-Square                     2      2.5961    0.2731

                         WARNING: 33% of the cells have expected counts less

                                  than 5. Chi-Square may not be a valid test.
Note the warning message in the output above, indicating that the chi-square test may not be appropriate, due to small expected counts in 33% of the cells of the table. In this case, you can use a Fisher’s exact test, by specifying the exact statement, as shown below:
/*For tables containing sparse cells, Fisher's exact test can be performed */

proc freq data=pulse;


tables sex * activity /nopercent nocol;


exact fisher;


where smokes=2;

run;
Selected output. Note: read the p-value for Fisher’s exact test from the output labeled Pr <=P, at the bottom of the output.
                                       Fisher's Exact Test

                                  ----------------------------------

                                  Table Probability (P)       0.0221

                                  Pr <= P                     0.2568

Compare mean of a continuous variable for a categorical variable with two levels:
If you wish to compare the means of a continuous variable between two independent groups, you can carry out an independent samples t test. Some of the assumptions for this test are:
1) The two groups are independent.

2) The variable whose means we are interested in should be roughly normally distributed within each group. You can do a Q-Q plot to check normality for each group.
3) The variability of the observations in each of the groups should be similar. A side-by-side boxplot is helpful to visually check the equal variance assumption. 
/* Association between a continuous variable and a categorical variable, FIRST, CHECK THE MEANS FOR EACH GROUP*/

proc means data=pulse;


class ran;


var pulse1 pulse2;

run;

/*Q-Q plot can be used to check normality assumption of test variables, at each level of the grouping variable */

proc sort data=pulse;


by ran;

run;

proc univariate data=pulse;


by ran;


var pulse1 pulse2;


histogram / normal;


qqplot /normal (mu=est sigma=est);

run;
Selected output:

Ran = “Yes”




Ran = “No”

/*Boxplot allows you to compare the test variables VISUALLY between two groups, for both location and spread */
proc sort data=pulse;

  by ran;
proc boxplot data=pulse;


plot (pulse1 pulse2) * ran;

run;

/*Indepedent samples t test*/

proc ttest data=pulse;


var pulse1 pulse2;


class ran;

run;
                                         The TTEST Procedure

                                             Statistics

                                Lower CL          Upper CL  Lower CL           Upper CL

   Variable  ran             N      Mean    Mean      Mean   Std Dev  Std Dev   Std Dev  Std Err

   pulse1    Yes            35    69.672    73.6    77.528    9.2501   11.436    14.983    1.933

   pulse1    No             57    69.551  72.421    75.291    9.1319   10.817    13.269   1.4327

   pulse1    Diff (1-2)           -3.537  1.1789    5.8951    9.6488   11.055    12.944   2.3739

   pulse2    Yes            35    86.007  92.514    99.022    15.323   18.943    24.819    3.202

   pulse2    No             57    69.676  72.316    74.955     8.399   9.9484    12.204   1.3177

   pulse2    Diff (1-2)           14.208  20.198    26.189    12.255   14.041     16.44   3.0152

                                               T-Tests

                Variable    Method           Variances      DF    t Value    Pr > |t|

                pulse1      Pooled           Equal          90       0.50      0.6207

                pulse1      Satterthwaite    Unequal        69       0.49      0.6257

                pulse2      Pooled           Equal          90       6.70      <.0001

                pulse2      Satterthwaite    Unequal      45.7       5.83      <.0001

                                        Equality of Variances

                    Variable    Method      Num DF    Den DF    F Value    Pr > F

                    pulse1      Folded F        34        56       1.12    0.6990

                    pulse2      Folded F        34        56       3.63    <.0001
If you are unwilling to assume normality for your test variable or the sample size is too small for you to appeal to the central-limit-theorem, you may want to use non-parametric tests, which are also known as distribution-free tests. Mann-Whitney/Wilcoxon test is the non-parametric analog of the independent sample t test.

/*Non-parametric test: Mann-Whitney/Wilcoxon test*/

proc npar1way data=pulse wilcoxon;


class ran;


var pulse1 pulse2;

run;
Selected output:

                                        The NPAR1WAY Procedure

                           Wilcoxon Scores (Rank Sums) for Variable pulse2

                                     Classified by Variable ran

                                    Sum of      Expected       Std Dev          Mean

                 ran       N        Scores      Under H0      Under H0         Score

                 -------------------------------------------------------------------

                 Yes      35        2292.0       1627.50    124.056919     65.485714

                 No       57        1986.0       2650.50    124.056919     34.842105

                                 Average scores were used for ties.

                                      Wilcoxon Two-Sample Test

                                   Statistic             2292.0000

                                   Normal Approximation

                                   Z                        5.3524

                                   One-Sided Pr >  Z        <.0001

                                   Two-Sided Pr > |Z|       <.0001

                                   t Approximation

                                   One-Sided Pr >  Z        <.0001

                                   Two-Sided Pr > |Z|       <.0001

                              Z includes a continuity correction of 0.5.

